An Alternative Approach in Formulating Diamatic Domes

H Nooshin and O A Samavati, 7th April 2015

Diamatic domes and their formulation, together with a normat (coordinate system) which is suitable for the formulation of diamatic domes are discussed in some length in FCP, Section 2.5 The acronym 'FCP' stands for 'Formex Configuration Processing', Part C of the book: 'Structural Morphology and Configuration Processing of Space Structures', Edited by R Motro, Multi-Science Publishing Co, ISBN: 978-1-907132-13-1, 2009.
The first three sections in Part C of this book are the reprinted versions of three papers by H Nooshin and P Disney, published in the 'International Journal of Space Structures':
Formex Configuration Processing I (Vol 15, No 1, 2000)
Formex Configuration Processing II (Vol 16, No 1, 2001)
Formex Configuration Processing III (Vol 17, No 1, 2002)

The purpose of the present document is to describe an alternative way of formulating the geometry of diamatic domes. The Reader is assumed to be well familiar with formex configuration processing and the material of Section 2.5 of FCP.



Figure 1. A simple diamatic dome.

To begin with, consider the diamatic dome of Fig 1. A parametric Formian scheme (program) for the formulation of the geometry of this dome is given as Scheme 1. This scheme follows the approach of formulation in Section 2.5 of FCP.

```
(*) Scheme 1: Formulation of the Diamatic Dome of Fig 1 (*)

S=36;      (*) Span of the dome (*)
H=8;        (*) Rise of the dome (*)
m=6;       (*) Frequency of pattern (*)
n=6;       (*) Number of sectors (*)
A=2*atan|(2*H/S);     (*) Sweep angle (*)
R=S/(2*sin|A);           (*) Circumradius of dome (*)

(*) E represents one sector of the dome relative to the diamatic normat coordinates (*)
E=genit(1,m,1,1,0,1)|{[1,0,0; 1,0,1], [1,0,0;1,1,1], [1,0,1; 1,1,1]};
F=bd(R, 360/n,A/m)|E;    (*) F is the same as E but in Cartesian coordinates (*)
Dome=pex|rosad(0,0,n,360/n)|F;

use &,vm(2),vt(2),lw(0.5),vh(R,R,3*R, 0,0,0, 0,0,1);
clear;   draw Dome;
<>
```

In Scheme 1, the parameters of the problem are initialised in lines 3 to 6. These parameters are the span of the dome, rise of the dome, frequency of the pattern and number of sectors of the dome. In lines 7 and 8, the sweep angle and the Circumradius of the dome are calculated using the relations given in Fig 1.9.2 of FCP.

Formex variable E, in line 11, represents the connectivity (compret) of one sector of the dome in terms of the diamatic normat coordinates, see FCP, Section 2.5. The instruction in line 12 creates the formex variable F that represents the same configuration as E but is given relative to the global Cartesian coordinate system. Finally, the instruction in line 13 creates the formex variable Dome that represents the required dome in the global Cartesian coordinate system. The instruction replicates the sector represented by F n times rotationally and removes the overlapping side elements using the ‘pex’ function.

The alternative approach for the formulation of the dome of Fig 1 concerns lines 12 and 13. To elaborate, the alternative formulation for lines 12 and 13 would be:

```
12   G=pex|lib(i=1,n)|glisat(0,0,1,0,-atan|i)|E;
13   Dome=bd(R,360/n,A/m)|G;
```

In this approach, the compret (connectivity) of the whole of the dome is generated in the diamatic normat coordinate system before transforming the information to the global Cartesian coordinates (in line 13). In the instruction of line 12, in addition to the ‘pex’ function, the ‘libra’ function is used for replication (see FCP, Section 2.A.12) as well as the shearing function ‘glisat’ (see FCP, Table 2.A.5).



Figure 2. Effect of glisat function in the diamatic normat

The glisat function usually has a ‘shearing’ effect. However, in the context of a diamatic normat, the glisat function has a rotational effect. For example, Fig 2 shows the first and second sectors of the dome of Fig 1. Also, shown shaded, are a panel of the first sector, denoted by T1, and the corresponding panel of the second sector, denoted by T2. Panels T1 and T2 have exactly the same shape and the difference between the shapes, that may be felt, is only due to the fact that Fig 2 is a perspective view.

Panels T1 and T2 in diamatic normat coordinates, may be written as:

```
T1=[1,0,4; 1,1,4; 1,1,5];
```
and
```
T2=glisat(0,0,1,0,-atan|1)|T1;
```

With these formulations, after the transfer of the normat coordinates to the global Cartesian coordinates, T2 is the rotation of T1 by 360°/n, about a vertical axis through the centre of the dome.